Correction du TP "Systèmes différentiels autonomes"

Exercice 1 On considère le système différentiel

$$\begin{cases} x'(t) = y(t) - x(t), \\ y'(t) = x(t)y(t) - 1. \end{cases} \tag{1}$$

1. Il s'agit d'un système différentiel autonome, i.e. de la forme $X'(t) = F(X(t))$, avec

$$X = \begin{pmatrix} x \\ y \end{pmatrix}, \text{ et } F\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} y - x \\ xy - 1 \end{pmatrix}.$$

D'après le théorème de Cauchy-Lipschitz, pour tout $X_0 \in \mathbb{R}^2$, ce système admet une unique solution maximale $X \in C^1(I, \mathbb{R}^2)$ telle que $X(0) = X_0$. De plus, I est ouvert. Les points stationnaires sont les solutions de $F(X) = 0$. Il y en a deux : $X^+ = (1, 1)$ et $X^- = (-1, -1)$. On a

$$DF(X) = \begin{pmatrix} -1 & 1 \\ y & x \end{pmatrix}.$$

En X^+, le système différentiel linéarisé s'écrit $\tilde{X}'(t) = A_+\tilde{X}(t)$, avec

$$A_+ = DF(X^+) = \begin{pmatrix} -1 & 1 \\ 1 & 1 \end{pmatrix}.$$

Le polynôme caractéristique de A^+ est $P_{A^+}(\lambda) = \lambda^2 - 2$. Les valeurs propres de A_+ sont $\pm\sqrt{2}$, donc on est en présence d'un point selle (ou col).

De même,

$$A_- = DF(X^-) = \begin{pmatrix} -1 & 1 \\ -1 & -1 \end{pmatrix}.$$

Le polynôme caractéristique de A^- est $P_{A^-}(\lambda) = \lambda^2 + 2\lambda + 2$. Les valeurs propres de A_- sont $-1 \pm i$. On est en présence d'un foyer attractif.

***

Questions 2-3-4

• **En Scilab**, sans utiliser `xclick`.

```
//
clear
deff("[xdot] = F(t,x)","xdot =[x(2)-x(1); x(1)*x(2)-1]")
xmin=-3;
xmax=3;
ymin=-3;
ymax=3;
xf= xmin:0.4:xmax;
yf= ymin:0.4:ymax;
clf
fchamp(F,0,xf,yf)
halt()
```

```
clf()
fchamp(F,0,xf,yf,1,[xmin,ymin,xmax,ymax],"011")
plot2d(xf,xf,style=1) //isocline verticale
xx1=[0.01:0.01:xmax];
plot2d(xx1,ones(xx1)./xx1,rect=[xmin,ymin,xmax,ymax])
xx2=[xmin:0.01:-0.01];
plot2d(xx2,ones(xx2)./xx2,rect=[xmin,ymin,xmax,ymax])

//conditions initiales pour le trace des trajectoires
x0=[-3 3;-3 -3; 0 -3;3 0.5; -2 3; 3 0.4;-1 3;3 0; -1.5 3]';
t=[0:0.01:5];
for j=1:size(x0,2)
j
    //calcul numerique de la courbe integrale de condition initiale x0(:,j)
    x=ode(x0(:,j),0,t,F);
    //trace de la trajectoire associee
    plot2d(x(1,:),x(2,:),style=2,rect=[xmin,ymin,xmax,ymax])
   end
```



Figure 1: Portrait de phase (Scilab)

Question 5 : On observe que l'allure du portrait de phase près de chacun des deux points stationnaires correspond à ce que l'on attend de l'étude du système linéarisé. Il faut noter que l'exécution du programme provoque pour certaines des conditions initiales un message d'avertissement.

• **Idem**, en utilisant `xclick`

Il faut cliquer 5 fois sur la figure graphique pour exécuter complètement le programme.

```
//un exemple d'utilisation de xclick
```

```
lines(0)
clear
deff("[xdot] = F(t,x)","xdot =[x(2)-x(1); x(1)*x(2)-1]")
xmin=-3;
xmax=3;
ymin=-3;
ymax=3;
xf= xmin:0.4:xmax;
yf= ymin:0.4:ymax;

clf()
fchamp(F,0,xf,yf,1,[xmin,ymin,xmax,ymax],"011")
plot2d(xf,xf,style=1) //isocline verticale
xx1=[0.01:0.01:xmax];
plot2d(xx1,ones(xx1)./xx1,rect=[xmin,ymin,xmax,ymax]) //isocline horizontale
xx2=[xmin:0.01:-0.01];
plot2d(xx2,ones(xx2)./xx2,rect=[xmin,ymin,xmax,ymax]) //isocline horizontale

//trace de trajectoires avec xclick
t=[0:0.01:5];
x0=zeros(2,1);
for j=1:5
    j
    //determiner la condition initiale avec xclick
    [n,x0(1),x0(2)]=xclick()
    //calcul numerique de la courbe integrale de condition initiale x0
    //NB : n ne nous sert a rien
    x=ode(x0,0,t,F);
    //trace de la trajectoire associee
    plot2d(x(1,:),x(2,:),style=2,rect=[xmin,ymin,xmax,ymax])
end
```

- **En Maple**

```
> with(DEtools):
> dfieldplot([diff(x(t),t)=y(t)-x(t),diff(y(t),t)=x(t)*y(t)-1],[x(t),y(t
> )],t=-2..2,x=-3..3,y=-3..3);

> with(plots):implicitplot({y-x=0,x*y-1=0},x=-3..3,y=-3..3);
# Warning, the name changecoords has been redefined

> eq1:=diff(x(t),t)=y-x:eq2:=diff(y(t),t)=x*y-1:
> phaseportrait([eq1,eq2],[x(t),y(t)],t=0..5,[[x(0)=-3,y(0)=3],[x(0)=-3,
> y(0)=-3],[x(0)=0,y(0)=-2.5],[x(0)=3,y(0)=0.5],[x(0)=-1.5,y(0)=3],[x(0)
> =3,y(0)=0.4]],x=-3..3,y=-3..3,stepsize=0.001,linecolor=blue);
# Warning, x is present as both a dependent variable and a name.
# Inconsistent specification of the dependent variable is deprecated,
# and it is assumed that the name is being used in place of the
# dependent variable.
# Warning, y is present as both a dependent variable and a name.
```

```
# Inconsistent specification of the dependent variable is deprecated,
# and it is assumed that the name is being used in place of the
# dependent variable.
```



Figure 2: Portrait de phase (Maple)